Однородные полиномиальные уравнения

В.Э. Адлер. Классические интегрируемые системы, весенний семестр 2020

Здесь излагается метод неопределенных коэффициентов для отыскания высших симметрий и законов сохранения, в случае уравнений

(1) $\quad u_t = u_k + P(u, u_1, \ldots, u_{k-1}),$

где в правой части стоит многочлен, однородный относительно веса $w(u_j) = m + n\,j,\quad m, n > 0.$ Для поиска симметрии вида $u_T = u_K + \ldots$ генерируем многочлен с той же однородностью и неизвестными коэффициентами. Вычисляем коммутатор, собираем коэффициенты, решаем линейную систему. Точно так же ищутся законы сохранения.

Определим общие функции для символьных вычислений, которые нам потребуются (они уже обсуждались в программе **02_Inversion of the Miura transformation.nb**)

In[1]:=

```
(* список переменных u[_] в выражении f *)
vars[f_, u_] := Union[Cases[f, _u, {0, ∞}]]

(* df, дифференциал f *)
dif[f_] := Plus @@ (D[f, #] dd[#] & /@ vars[f, u])

(* Dx(f), Dx^k(f), полная производная по x от f *)
dx[f_] := D[f, x] + dif[f] /. dd[u[k_]] :> u[k + 1]
dx[f_, k_] := Nest[dx, f, k]

(* g*(f), оператор линеаризации g, примененный к f *)
star[g_, f_] := dif[g] /. dd[u[k_]] :> dx[f, k]

(* [f,g]=g*(f)-f*(g) *)
comm[f_, g_] := Expand[star[g, f] - star[f, g]]

(* E(f), вариационная производная или оператор Эйлера *)
vard[f_] := Expand[Plus @@ ((-1)^#[[1]] dx[D[f, #], #[[1]]] & /@ vars[f, u])]
```

In[8]:=
```
(* порядок по производным *)
ord[f_] := Module[{v = vars[f, u]}, If[v == {}, -1, v[[-1]][[1]]]]

(* интегрирование по частям *)
int[f_] := Module[{a = 0, b = Expand[f], c, k},
  While[k = ord[b];   (* вычисляем k – порядок b *)
   And[k > 0,          (* выполняем цикл, пока выполняется условие, что k>0 и *)
    c = D[b, u[k]];   (* b линейна по старшей переменной *)
    Expand[D[c, u[k]]] == 0
   ],
   c = Integrate[c, u[k - 1]];  (* интегрируем по переменной, следующей по старшинству *)
   a = Expand[a + c];           (* пересчет a и b *)
   b = Expand[b - dx[c]]
  ];
  If[ord[b] == -1,   (* если в результате остаток не зависит от u, то это функция от x *)
   {Expand[a + Integrate[b, x]], 0},
   {a, b}
  ]
 ]
```

Дополнительно заготовим несколько функций для генерации многочленов:

mon[*d*, *r*, *k*] список мономов степени d от u_k, u_{k+1}, ... с общей суммой индексов равной r;

hom[*m*, *n*, *M*] список всех мономов заданного веса M относительно веса $w(u_j) = m + j\,n$;

rhom[*m*, *n*, *M*] то же самое, но без мономов, линейных по старшей переменной (следовательно, все мономы в этом списке неэквивалентны по модулю $\operatorname{Im} D_x$);

pol[*l*, *c*] для заданного списка l длины n составляет линейную комбинацию $c[1]\,l[[1]] + \ldots + c[n]\,l[[n]]$;

cfs[*p*, *u*] список коэффициентов при всех мономах в многочлене p по переменным u. Повторяющиеся коэффициенты отбрасываются. Результат такой же, как от варианта с CoefficientList, но гораздо быстрее, если число переменных велико.

In[10]:=
```
mon[d_, r_, k_] := If[d == 1,
  If[r ≥ k, {u[r]}, {}],
  Flatten[Table[u[s] mon[d - 1, r - s, s], {s, k, r}]]
 ]

hom[m_, n_, M_] := Flatten[
  Table[
   If[IntegerQ[(M - m d) / n], mon[d, (M - m d) / n, 0], {}],
   {d, 1, Floor[M / m]}]
 ]

rhom[m_, n_, M_] := Select[hom[m, n, M], Exponent[#, Last[vars[#, u]]] > 1 || # == u[0] &]

pol[l_, c_] := c /@ Range[Length[l]].l

cfs[p_, u_] := Module[{z, pp = Expand[p]},
  If[FreeQ[pp, Plus], {pp /. _u → 1},
   Union[
    Reap[Map[Sow[z = # /. _u → 1, # / z] &, pp];, _, Plus @@ #2 &][[2]]
   ]]]
```

Тест. Сгенерируем длинный список, превратим его в многочлен, потом выпишем коэффициенты. Сделаем то же самое при помощи CoefficientList и сравним быстродействие.

In[15]:=
```
hom[2, 1, 15];
Length[%]
A = pol[%%, c]

cfs[A, u]
Complement[Flatten[CoefficientList[A, vars[A, u]]], {0}]

Timing[Do[ClearSystemCache[];
  cfs[A, u],
  {k, 1, 50}]]

Timing[Do[ClearSystemCache[];
  Complement[Flatten[CoefficientList[A, vars[A, u]]], {0}],
  {k, 1, 50}]]
```

Out[16]= 41

Out[17]=
```
c[41] u[0]^6 u[1] + c[40] u[0]^3 u[1]^3 + c[37] u[1]^5 + c[39] u[0]^4 u[1] u[2] +
 c[36] u[0] u[1]^3 u[2] + c[35] u[0]^2 u[1] u[2]^2 + c[30] u[1] u[2]^3 + c[38] u[0]^5 u[3] +
 c[34] u[0]^2 u[1]^2 u[3] + c[33] u[0]^3 u[2] u[3] + c[29] u[1]^2 u[2] u[3] + c[27] u[0] u[2]^2 u[3] +
 c[26] u[0] u[1] u[3]^2 + c[19] u[3]^3 + c[32] u[0]^3 u[1] u[4] + c[28] u[1]^3 u[4] +
 c[25] u[0] u[1] u[2] u[4] + c[23] u[0]^2 u[3] u[4] + c[18] u[2] u[3] u[4] + c[16] u[1] u[4]^2 +
 c[31] u[0]^4 u[5] + c[24] u[0] u[1]^2 u[5] + c[22] u[0]^2 u[2] u[5] + c[17] u[2]^2 u[5] +
 c[15] u[1] u[3] u[5] + c[12] u[0] u[4] u[5] + c[21] u[0]^2 u[1] u[6] + c[14] u[1] u[2] u[6] +
 c[11] u[0] u[3] u[6] + c[7] u[5] u[6] + c[20] u[0]^3 u[7] + c[13] u[1]^2 u[7] +
 c[10] u[0] u[2] u[7] + c[6] u[4] u[7] + c[9] u[0] u[1] u[8] + c[5] u[3] u[8] +
 c[8] u[0]^2 u[9] + c[4] u[2] u[9] + c[3] u[1] u[10] + c[2] u[0] u[11] + c[1] u[13]
```

Out[18]=
```
{c[1], c[2], c[3], c[4], c[5], c[6], c[7], c[8], c[9], c[10], c[11],
 c[12], c[13], c[14], c[15], c[16], c[17], c[18], c[19], c[20], c[21],
 c[22], c[23], c[24], c[25], c[26], c[27], c[28], c[29], c[30], c[31],
 c[32], c[33], c[34], c[35], c[36], c[37], c[38], c[39], c[40], c[41]}
```

Out[19]=
```
{c[1], c[2], c[3], c[4], c[5], c[6], c[7], c[8], c[9], c[10], c[11],
 c[12], c[13], c[14], c[15], c[16], c[17], c[18], c[19], c[20], c[21],
 c[22], c[23], c[24], c[25], c[26], c[27], c[28], c[29], c[30], c[31],
 c[32], c[33], c[34], c[35], c[36], c[37], c[38], c[39], c[40], c[41]}
```

Out[20]= {0.03125, Null}

Out[21]= {2.1875, Null}

Определим функцию **sym[ut, *m*, *n*, *K*]**, возвращающую, для заданной правой части u_t, правую часть симметрии $u_T = u_K + ...$, с однородностью $w(u_j) = m + n\, j$. Однородность u_t не проверяется, хотя это несложно сделать.

Для поиска законов сохранения поступаем аналогично, но используем rhom вместо hom. Дифференцируем многочлен ρ по t, применяем оператор Эйлера, приравниваем нулю и решаем систему уравнений на коэффициенты (альтернативно, можно не применять оператор Эйлера, а решать систему $D_t(\rho) = D_x(\sigma)$ в которой и ρ и σ – многочлены с неопределёнными коэффициентами).

In[22]:=
```
sym[ut_, m_, n_, K_] := Module[{uT, s, cc, c},
  (* однородный многочлен с коэффициентами c[j] *)
  uT = pol[hom[m, n, m + K n], c];
  (* коэффициент при u_K делаем равным 1 *)
  uT = uT /. Coefficient[uT, u[K]] → 1;
  (* список неопределенных коэффициентов *)
  cc = Complement[cfs[uT, u], {1}];
  (* вычисляем коммутатор, собираем коэффициенты, решаем систему уравнений на них *)
  s = Solve[cfs[comm[uT, ut], u] == 0, cc];
  (* если решения нет, возвращаем 0 *)
  If[s == {}, 0, Expand[uT /. s[[1]]]]
 ]

density[ut_, m_, n_, M_] := Module[{rho, s, c},
  rho = pol[rhom[m, n, M], c] /. c[1] → 1;
  s = Solve[cfs[vard[star[rho, ut]], u] == 0];
  If[s == {}, 0, Expand[rho /. s[[1]]]]
 ]
```

В **density** мы полагаем коэффициент при старшем члене равным 1, чтобы решение определялось однозначно. В принципе, это может привести к потере ответа, если этот коэффициент на самом деле равен 0, но в тех примерах, что мы будем рассматривать, все оказывается в порядке.

1 Вычисление симметрий уравнения Бюргерса

Уравнение Бюргерса

$$u_t = u_2 + 2\, u\, u_1$$

однородно относительно веса $w(u_j) = 1 + j$. У него есть высшие симметрии любого порядка. Зато, нет плотностей, кроме u (можно доказать, что уравнение порядка $2\,k$ не может иметь плотностей порядка выше k).

In[24]:=
```
M = 8;
Clear[ut]

Do[ut[k] = sym[u[2] + 2 u[0] u[1], 1, 1, k], {k, 1, M}]
```

Для красоты выводим симметрии в индексной записи.

In[27]:=
```
Do[Print[ut[j] /. u[n_] :→ u_n], {j, 1, M}]
```

u_1

$2 u_0 u_1 + u_2$

$3 u_0^2 u_1 + 3 u_1^2 + 3 u_0 u_2 + u_3$

$4 u_0^3 u_1 + 12 u_0 u_1^2 + 6 u_0^2 u_2 + 10 u_1 u_2 + 4 u_0 u_3 + u_4$

$5 u_0^4 u_1 + 30 u_0^2 u_1^2 + 15 u_1^3 + 10 u_0^3 u_2 + 50 u_0 u_1 u_2 + 10 u_2^2 + 10 u_0^2 u_3 + 15 u_1 u_3 + 5 u_0 u_4 + u_5$

$6 u_0^5 u_1 + 60 u_0^3 u_1^2 + 90 u_0 u_1^3 + 15 u_0^4 u_2 + 150 u_0^2 u_1 u_2 + 105 u_1^2 u_2 + 60 u_0 u_2^2 + 20 u_0^3 u_3 + 90 u_0 u_1 u_3 + 35 u_2 u_3 + 15 u_0^2 u_4 + 21 u_1 u_4 + 6 u_0 u_5 + u_6$

$7 u_0^6 u_1 + 105 u_0^4 u_1^2 + 315 u_0^2 u_1^3 + 105 u_1^4 + 21 u_0^5 u_2 + 350 u_0^3 u_1 u_2 + 735 u_0 u_1^2 u_2 + 210 u_0^2 u_2^2 + 280 u_1 u_2^2 + 35 u_0^4 u_3 + 315 u_0^2 u_1 u_3 + 210 u_1^2 u_3 + 245 u_0 u_2 u_3 + 35 u_3^2 + 35 u_0^3 u_4 + 147 u_0 u_1 u_4 + 56 u_2 u_4 + 21 u_0^2 u_5 + 28 u_1 u_5 + 7 u_0 u_6 + u_7$

$8 u_0^7 u_1 + 168 u_0^5 u_1^2 + 840 u_0^3 u_1^3 + 840 u_0 u_1^4 + 28 u_0^6 u_2 + 700 u_0^4 u_1 u_2 + 2940 u_0^2 u_1^2 u_2 + 1260 u_1^3 u_2 + 560 u_0^3 u_2^2 + 2240 u_0 u_1 u_2^2 + 280 u_2^3 + 56 u_0^5 u_3 + 840 u_0^3 u_1 u_3 + 1680 u_0 u_1^2 u_3 + 980 u_0^2 u_2 u_3 + 1260 u_1 u_2 u_3 + 280 u_0 u_3^2 + 70 u_0^4 u_4 + 588 u_0^2 u_1 u_4 + 378 u_1^2 u_4 + 448 u_0 u_2 u_4 + 126 u_3 u_4 + 56 u_0^3 u_5 + 224 u_0 u_1 u_5 + 84 u_2 u_5 + 28 u_0^2 u_6 + 36 u_1 u_6 + 8 u_0 u_7 + u_8$

Проверим, что симметрии попарно коммутируют.

In[28]:=
```
Table[comm[ut[i], ut[j]], {i, 1, M - 1}, {j, i + 1, M}]
```

Out[28]=
```
{{0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0}, {0, 0}, {0}}
```

Плотностей кроме u нет.

In[29]:=
```
Table[density[ut[2], 1, 1, k], {k, 1, 5}]
```

Out[29]=
```
{u[0], 0, 0, 0, 0}
```

2 Вычисление симметрий КдФ

Уравнение КдФ

$$u_t = u_3 + 6 u u_1$$

имеет вес $w(u_j) = 2 + j$. Вычисляем его симметрии до некоторого порядка. Можно убедиться, что симметрий четного порядка нет.

In[30]:=
```
M = 15;
Clear[ut]

Timing[Do[ut[k] = sym[u[3] + 6 u[0] u[1], 2, 1, k], {k, 1, M}]]

Do[Print[ut[j] /. u[n_] :> u_n], {j, 1, M}]
```

Out[32]=
```
{0.453125, Null}
```

u_1

0

$6\, u_0\, u_1 + u_3$

0

$30\, u_0^2\, u_1 + 20\, u_1\, u_2 + 10\, u_0\, u_3 + u_5$

0

$140\, u_0^3\, u_1 + 70\, u_1^3 + 280\, u_0\, u_1\, u_2 + 70\, u_0^2\, u_3 + 70\, u_2\, u_3 + 42\, u_1\, u_4 + 14\, u_0\, u_5 + u_7$

0

$$630\, u_0^4\, u_1 + 1260\, u_0\, u_1^3 + 2520\, u_0^2\, u_1\, u_2 + 1302\, u_1\, u_2^2 + 420\, u_0^3\, u_3 + 966\, u_1^2\, u_3 + 1260\, u_0\, u_2\, u_3 + 756\, u_0\, u_1\, u_4 + 252\, u_3\, u_4 + 126\, u_0^2\, u_5 + 168\, u_2\, u_5 + 72\, u_1\, u_6 + 18\, u_0\, u_7 + u_9$$

0

$$2772\, u_0^5\, u_1 + 13\,860\, u_0^2\, u_1^3 + 18\,480\, u_0^3\, u_1\, u_2 + 14\,784\, u_1^3\, u_2 + 28\,644\, u_0\, u_1\, u_2^2 + 2310\, u_0^4\, u_3 + 21\,252\, u_0\, u_1^2\, u_3 + 13\,860\, u_0^2\, u_2\, u_3 + 9702\, u_2^2\, u_3 + 7194\, u_1\, u_3^2 + 8316\, u_0^2\, u_1\, u_4 + 11\,484\, u_1\, u_2\, u_4 + 5544\, u_0\, u_3\, u_4 + 924\, u_0^3\, u_5 + 2838\, u_1^2\, u_5 + 3696\, u_0\, u_2\, u_5 + 924\, u_4\, u_5 + 1584\, u_0\, u_1\, u_6 + 660\, u_3\, u_6 + 198\, u_0^2\, u_7 + 330\, u_2\, u_7 + 110\, u_1\, u_8 + 22\, u_0\, u_9 + u_{11}$$

0

$$12\,012\, u_0^6\, u_1 + 120\,120\, u_0^3\, u_1^3 + 30\,030\, u_1^5 + 120\,120\, u_0^4\, u_1\, u_2 + 384\,384\, u_0\, u_1^3\, u_2 + 372\,372\, u_0^2\, u_1\, u_2^2 + 177\,320\, u_1\, u_2^3 + 12\,012\, u_0^5\, u_3 + 276\,276\, u_0^2\, u_1^2\, u_3 + 120\,120\, u_0^3\, u_2\, u_3 + 392\,964\, u_1^2\, u_2\, u_3 + 252\,252\, u_0\, u_2^2\, u_3 + 187\,044\, u_0\, u_1\, u_3^2 + 26\,598\, u_3^3 + 72\,072\, u_0^3\, u_1\, u_4 + 77\,220\, u_1^3\, u_4 + 298\,584\, u_0\, u_1\, u_2\, u_4 + 72\,072\, u_0^2\, u_3\, u_4 + 126\,984\, u_2\, u_3\, u_4 + 37\,466\, u_1\, u_4^2 + 6006\, u_0^4\, u_5 + 73\,788\, u_0\, u_1^2\, u_5 + 48\,048\, u_0^2\, u_2\, u_5 + 42\,042\, u_2^2\, u_5 + 62\,348\, u_1\, u_3\, u_5 + 24\,024\, u_0\, u_4\, u_5 + 20\,592\, u_0^2\, u_1\, u_6 + 35\,464\, u_1\, u_2\, u_6 + 17\,160\, u_0\, u_3\, u_6 + 3432\, u_5\, u_6 + 1716\, u_0^3\, u_7 + 6578\, u_1^2\, u_7 + 8580\, u_0\, u_2\, u_7 + 2574\, u_4\, u_7 + 2860\, u_0\, u_1\, u_8 + 1430\, u_3\, u_8 + 286\, u_0^2\, u_9 + 572\, u_2\, u_9 + 156\, u_1\, u_{10} + 26\, u_0\, u_{11} + u_{13}$$

0

$$51\,480\, u_0^7\, u_1 + 900\,900\, u_0^4\, u_1^3 + 900\,900\, u_0\, u_1^5 + 720\,720\, u_0^5\, u_1\, u_2 + 5\,765\,760\, u_0^2\, u_1^3\, u_2 + 3\,723\,720\, u_0^3\, u_1\, u_2^2 + 4\,178\,460\, u_1^3\, u_2^2 + 5\,319\,600\, u_0\, u_1\, u_2^3 + 60\,060\, u_0^6\, u_3 + 2\,762\,760\, u_0^3\, u_1^2\, u_3 + 1\,540\,110\, u_1^4\, u_3 + 900\,900\, u_0^4\, u_2\, u_3 + 11\,788\,920\, u_0\, u_1^2\, u_2\, u_3 + 3\,783\,780\, u_0^2\, u_2^2\, u_3 + 2\,279\,420\, u_2^3\, u_3 + 2\,805\,660\, u_0^2\, u_1\, u_3^2 + 5\,045\,040\, u_1\, u_2\, u_3^2 + 797\,940\, u_0\, u_3^3 + 540\,540\, u_0^4\, u_1\, u_4 + 2\,316\,600\, u_0\, u_1^3\, u_4 + 4\,478\,760\, u_0^2\, u_1\, u_2\, u_4 + 4\,010\,292\, u_1\, u_2^2\, u_4 + 720\,720\, u_0^3\, u_3\, u_4 + 2\,961\,816\, u_1^2\, u_3\, u_4 + 3\,809\,520\, u_0\, u_2\, u_3\, u_4 + 1\,123\,980\, u_0\, u_1\, u_4^2 + 578\,006\, u_3\, u_4^2 + 36\,036\, u_0^5\, u_5 + 1\,106\,820\, u_0^2\, u_1^2\, u_5 + 480\,480\, u_0^3\, u_2\, u_5 + 1\,959\,672\, u_1^2\, u_2\, u_5 + 1\,261\,260\, u_0\, u_2^2\, u_5 + 1\,870\,440\, u_0\, u_1\, u_3\, u_5 + 480\,194\, u_3^2\, u_5 + 360\,360\, u_0^2\, u_4\, u_5 + 763\,620\, u_2\, u_4\, u_5 + 187\,174\, u_1\, u_5^2 + 205\,920\, u_0^3\, u_1\, u_6 + 274\,560\, u_1^3\, u_6 + 1\,063\,920\, u_0\, u_1\, u_2\, u_6 + 257\,400\, u_0^2\, u_3\, u_6 + 543\,400\, u_2\, u_3\, u_6 + 320\,632\, u_1\, u_4\, u_6 + 102\,960\, u_0\, u_5\, u_6 + 12\,870\, u_0^4\, u_7 + 197\,340\, u_0\, u_1^2\, u_7 + 128\,700\, u_0^2\, u_2\, u_7 + 134\,706\, u_2^2\, u_7 + 199\,888\, u_1\, u_3\, u_7 + 77\,220\, u_0\, u_4\, u_7 + 12\,870\, u_6\, u_7 + 42\,900\, u_0^2\, u_1\, u_8 + 88\,400\, u_1\, u_2\, u_8 + 42\,900\, u_0\, u_3\, u_8 + 10\,010\, u_5\, u_8 + 2860\, u_0^3\, u_9 + 13\,130\, u_1^2\, u_9 + 17\,160\, u_0\, u_2\, u_9 + 6006\, u_4\, u_9 + 4680\, u_0\, u_1\, u_{10} + 2730\, u_3\, u_{10} + 390\, u_0^2\, u_{11} + 910\, u_2\, u_{11} + 210\, u_1\, u_{12} + 30\, u_0\, u_{13} + u_{15}$$

Проверим, что симметрии попарно коммутируют, ограничившись порядком 11.

In[34]:=
```
Table[comm[ut[i], ut[j]], {i, 1, 9, 2}, {j, i + 2, 11, 2}]
```

Out[34]=
```
{{0, 0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0}, {0, 0}, {0}}
```

3 Вычисление плотностей для законов сохранения КдФ

Вот, для примера, все однородные мономы веса 22, неэквивалентные по модулю $\operatorname{Im} D_x$

In[35]:=
```
rhom[2, 1, 22]
Length[%]
```

Out[35]= $\{u[9]^2, u[0]\,u[8]^2, u[2]\,u[7]^2, u[4]\,u[6]^2, u[0]^2\,u[7]^2, u[0]\,u[2]\,u[6]^2,$
$u[0]\,u[4]\,u[5]^2, u[1]^2\,u[6]^2, u[1]\,u[3]\,u[5]^2, u[2]^2\,u[5]^2, u[2]\,u[4]^3, u[3]^2\,u[4]^2,$
$u[0]^3\,u[6]^2, u[0]^2\,u[2]\,u[5]^2, u[0]^2\,u[4]^3, u[0]\,u[1]^2\,u[5]^2, u[0]\,u[1]\,u[3]\,u[4]^2,$
$u[0]\,u[2]^2\,u[4]^2, u[0]\,u[3]^4, u[1]^2\,u[2]\,u[4]^2, u[1]\,u[2]\,u[3]^3, u[2]^3\,u[3]^2,$
$u[0]^4\,u[5]^2, u[0]^3\,u[2]\,u[4]^2, u[0]^2\,u[1]^2\,u[4]^2, u[0]^2\,u[1]\,u[3]^3, u[0]^2\,u[2]^2\,u[3]^2,$
$u[0]\,u[1]^2\,u[2]\,u[3]^2, u[0]\,u[2]^5, u[1]^4\,u[3]^2, u[1]^2\,u[2]^4, u[0]^5\,u[4]^2, u[0]^4\,u[2]\,u[3]^2,$
$u[0]^3\,u[1]^2\,u[3]^2, u[0]^3\,u[2]^4, u[0]^2\,u[1]^2\,u[2]^3, u[0]\,u[1]^4\,u[2]^2, u[0]^6\,u[3]^2,$
$u[0]^5\,u[2]^3, u[0]^4\,u[1]^2\,u[2]^2, u[0]^2\,u[1]^6, u[0]^7\,u[2]^2, u[0]^5\,u[1]^4, u[0]^8\,u[1]^2, u[0]^{11}\}$

Out[36]= 45

Именно до этого веса дошёл Миура-сэнсэй, только без компьютера, но не осилил; и тогда он придумал своё преобразование.

In[37]:=
```
ut = u[3] + 6 u[0] u[1];
Do[rho[j] = density[ut, 2, 1, j], {j, 4, 22, 2}]

(* проверка *)
Table[vard[star[rho[j], ut]], {j, 4, 22, 2}]

(* вывод в индексной форме *)
Do[Print[rho[j] /. u[n_] :> u_n], {j, 4, 22, 2}]
```

Out[39]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

$$u_0^2$$

$$-2\,u_0^3+u_1^2$$

$$5\,u_0^4-10\,u_0\,u_1^2+u_2^2$$

$$-14\,u_0^5+70\,u_0^2\,u_1^2-14\,u_0\,u_2^2+u_3^2$$

$$42\,u_0^6-420\,u_0^3\,u_1^2-35\,u_1^4+126\,u_0^2\,u_2^2+20\,u_2^3-18\,u_0\,u_3^2+u_4^2$$

$$-132\,u_0^7+2310\,u_0^4\,u_1^2+770\,u_0\,u_1^4-924\,u_0^3\,u_2^2-462\,u_1^2\,u_2^2-440\,u_0\,u_2^3+198\,u_0^2\,u_3^2+110\,u_2\,u_3^2-22\,u_0\,u_4^2+u_5^2$$

$$429\,u_0^8-12\,012\,u_0^5\,u_1^2-10\,010\,u_0^2\,u_1^4+6006\,u_0^4\,u_2^2+12\,012\,u_0\,u_1^2\,u_2^2+5720\,u_0^2\,u_2^3+1001\,u_2^4-1716\,u_0^3\,u_3^2-858\,u_1^2\,u_3^2-2860\,u_0\,u_2\,u_3^2+286\,u_0^2\,u_4^2+182\,u_2\,u_4^2-26\,u_0\,u_5^2+u_6^2$$

$$-1430\,u_0^9+60\,060\,u_0^6\,u_1^2+100\,100\,u_0^3\,u_1^4+10\,010\,u_1^6-36\,036\,u_0^5\,u_2^2-180\,180\,u_0^2\,u_1^2\,u_2^2-57\,200\,u_0^3\,u_2^3-52\,624\,u_1^2\,u_2^3-30\,030\,u_0\,u_2^4+12\,870\,u_0^4\,u_3^2+25\,740\,u_0\,u_1^2\,u_3^2+42\,900\,u_0^2\,u_2\,u_3^2+16\,406\,u_2^2\,u_3^2+5096\,u_1\,u_3^3-2860\,u_0^3\,u_4^2-1430\,u_1^2\,u_4^2-5460\,u_0\,u_2\,u_4^2-252\,u_4^3+390\,u_0^2\,u_5^2+280\,u_2\,u_5^2-30\,u_0\,u_6^2+u_7^2$$

$$4862\,u_0^{10}-291\,720\,u_0^7\,u_1^2-850\,850\,u_0^4\,u_1^4-340\,340\,u_0\,u_1^6+204\,204\,u_0^6\,u_2^2+2\,042\,040\,u_0^3\,u_1^2\,u_2^2+510\,510\,u_1^4\,u_2^2+486\,200\,u_0^4\,u_2^3+1\,789\,216\,u_0\,u_1^2\,u_2^3+510\,510\,u_0^2\,u_2^4+101\,660\,u_2^5-87\,516\,u_0^5\,u_3^2-437\,580\,u_0^2\,u_1^2\,u_3^2-486\,200\,u_0^3\,u_2\,u_3^2-418\,132\,u_1^2\,u_2\,u_3^2-557\,804\,u_0\,u_2^2\,u_3^2-173\,264\,u_0\,u_1\,u_3^3-16\,099\,u_3^4+24\,310\,u_0^4\,u_4^2+48\,620\,u_0\,u_1^2\,u_4^2+92\,820\,u_0^2\,u_2\,u_4^2+38\,726\,u_2^2\,u_4^2+32\,368\,u_1\,u_3\,u_4^2+8568\,u_0\,u_4^3-4420\,u_0^3\,u_5^2-2210\,u_1^2\,u_5^2-9520\,u_0\,u_2\,u_5^2-1428\,u_4\,u_5^2+510\,u_0^2\,u_6^2+408\,u_2\,u_6^2-34\,u_0\,u_7^2+u_8^2$$

$$-16\,796\,u_0^{11}+1\,385\,670\,u_0^8\,u_1^2+6\,466\,460\,u_0^5\,u_1^4+6\,466\,460\,u_0^2\,u_1^6-1\,108\,536\,u_0^7\,u_2^2-19\,399\,380\,u_0^4\,u_1^2\,u_2^2-19\,399\,380\,u_0\,u_1^4\,u_2^2-3\,695\,120\,u_0^5\,u_2^3-33\,995\,104\,u_0^2\,u_1^2\,u_2^3-6\,466\,460\,u_0^3\,u_2^4-10\,069\,202\,u_1^2\,u_2^4-3\,863\,080\,u_0\,u_2^5+554\,268\,u_0^6\,u_3^2+5\,542\,680\,u_0^3\,u_1^2\,u_3^2+1\,385\,670\,u_1^4\,u_3^2+4\,618\,900\,u_0^4\,u_2\,u_3^2+15\,889\,016\,u_0\,u_1^2\,u_2\,u_3^2+10\,598\,276\,u_0^2\,u_2^2\,u_3^2+3\,649\,900\,u_2^3\,u_3^2+3\,292\,016\,u_0^2\,u_1\,u_3^3+\frac{11\,795\,960}{3}\,u_1\,u_2\,u_3^3+611\,762\,u_0\,u_3^4-184\,756\,u_0^5\,u_4^2-923\,780\,u_0^2\,u_1^2\,u_4^2-1\,175\,720\,u_0^3\,u_2\,u_4^2-957\,372\,u_1^2\,u_2\,u_4^2-1\,471\,588\,u_0\,u_2^2\,u_4^2-1\,229\,984\,u_0\,u_1\,u_3\,u_4^2-251\,294\,u_3^2\,u_4^2-162\,792\,u_0^2\,u_4^3-173\,128\,u_2\,u_4^3+41\,990\,u_0^4\,u_5^2+83\,980\,u_0\,u_1^2\,u_5^2+180\,880\,u_0^2\,u_2\,u_5^2+82\,042\,u_2^2\,u_5^2+62\,016\,u_1\,u_3\,u_5^2+54\,264\,u_0\,u_4\,u_5^2-6460\,u_0^3\,u_6^2-3230\,u_1^2\,u_6^2-15\,504\,u_0\,u_2\,u_6^2-2508\,u_4\,u_6^2+646\,u_0^2\,u_7^2+570\,u_2\,u_7^2-38\,u_0\,u_8^2+u_9^2$$

Можно заметить, что полученные плотности не совпадают с теми, что получаются по рекуррентной схеме (см. программу 02_Inversion of the Miura transformation.nb), но это не удивительно, так как сейчас мы убрали все члены, эквивалентные по модулю $\operatorname{Im} D_x$.

4 Поиск новых примеров

До сих пор мы строили симметрии для наперёд заданного уравнения. Теперь попробуем искать одновременно уравнение и симметрию. Это резко усложняет задачу. Действительно, если уравнение задано, то определяющая система для неизвестных коэффициентов линейная. Если же и в самом уравнении коэффициенты неизвестны, то получается квадратичная система. Тем не менее, при не очень высоких порядках её удаётся решить и это даёт несколько любопытных уравнений 3-го и 5-го порядков, обладающих симметриями.

Сейчас мы воспроизведём эти примеры. Мы ограничимся весами $w(u_j) = 2 + j$ (такой же как у КдФ) и $w(u_j) = 1 + 2\,j$. Имеются также несколько уравнений с весом $w(u_j) = 1 + j$, но они сводятся подстановками к уже найденным и мы пропустим этот случай. В других весах ответ оказывается пустым, это мы тоже не будем проверять. Также есть несколько полиномиальных уравнений, у которых коэффициент при старшей производной непостоянный, например, уравнение Дима $u_t = u^3\,u_3$. Они тоже сводятся, более сложными заменами, к уже найденным.

Все эти примеры были найдены в конце 70-х. Оказывается, что кроме них других однородных полиномиальных иерархий с нет. Это было доказано сравнительно недавно. Задача классификации интегрируемых уравнений с правой частью общего вида не решена полностью даже для уравнений третьего порядка $u_t = f(x, u, u_1, u_2, u_3)$, не говоря о пятом.

4.1 вес $w(u_j) = 2 + j$

В весе 5 имеется лишь два монома, так что кроме КдФ ответов нет.

Вес 6. Коэффициент при u_4 пусть равен 1. Функция sym теперь не годится. Формируем симметрию порядка 5, выписываем уравнения на коэффициенты и пытаемся их решить. Функция GroebnerBasis помогает упростить систему, дальше применяем Solve. Нашелся только нулевой ответ, то есть линейные уравнения $u_t = u_4$, $u_T = u_5$, что не интересно.

In[41]:=
```
hom[2, 1, 6]
ut = pol[%, a] /. a[1] → 1
uT = pol[hom[2, 1, 7], b] /. b[1] → 1

eqs = cfs[comm[uT, ut], u]
coef = Union[vars[ut, a], vars[uT, b]]

GroebnerBasis[eqs, coef]
sol = Solve[% == 0]
```

Out[41]= $\{u[4], u[0]\, u[2], u[1]^2, u[0]^3\}$

Out[42]= $a[4]\, u[0]^3 + a[3]\, u[1]^2 + a[2]\, u[0]\, u[2] + u[4]$

Out[43]= $b[4]\, u[0]^2 u[1] + b[3]\, u[1]\, u[2] + b[2]\, u[0]\, u[3] + u[5]$

Out[44]= {-a[4] b[2], -5 a[2] + 4 b[2], -10 a[2] - 10 a[3] + 6 b[2] + 4 b[3],
 -15 a[2] - 20 a[3] + 4 b[2] + 10 b[3], -2 a[4] b[4],
 -30 a[4] - a[2] b[2] + 8 b[4], -60 a[4] + a[3] b[2] - 2 a[2] b[3] + 20 b[4],
 -60 a[4] - 3 a[2] b[2] - 6 a[3] b[2] + 2 a[2] b[3] + 20 b[4],
 -90 a[4] - a[2] b[3] - 2 a[3] b[3] + 30 b[4], -18 a[4] b[2] - 3 a[4] b[3] + 4 a[2] b[4],
 -6 a[4] b[2] - 6 a[4] b[3] + 2 a[2] b[4] + 2 a[3] b[4]}

Out[45]= {a[2], a[3], a[4], b[2], b[3], b[4]}

Out[46]= $\{b[4], b[3]^2, 2\, b[2] - b[3], a[4], 10\, a[3] - 3\, b[3], 5\, a[2] - 2\, b[3]\}$

Out[47]= {{a[2] → 0, a[3] → 0, a[4] → 0, b[2] → 0, b[3] → 0, b[4] → 0}}

Попробуем, для того же u_t, взять симметрию порядка 6. Опять пусто.

In[48]:=
```
uT = pol[hom[2, 1, 8], b] /. b[1] → 1

eqs = cfs[comm[uT, ut], u];
coef = Union[vars[ut, a], vars[uT, b]];

GroebnerBasis[eqs, coef];
sol = Solve[% == 0]
```

Out[48]= $b[7]\, u[0]^4 + b[6]\, u[0]\, u[1]^2 + b[5]\, u[0]^2 u[2] + b[4]\, u[2]^2 + b[3]\, u[1]\, u[3] + b[2]\, u[0]\, u[4] + u[6]$

Out[52]= {{a[2] → 0, a[3] → 0, a[4] → 0, b[2] → 0, b[3] → 0, b[4] → 0, b[5] → 0, b[6] → 0, b[7] → 0}}

Вес 7. Коэффициент при u_5 пусть равен 1, а при $u\, u_3$ пусть равен 5 (этим мы исключаем линейное уравнение):

$$u_t = u_5 + 5\, u\, u_3 + \ldots$$

Симметрию возьмем веса 9, $u_\tau = u_7 + ...$

In[53]:=
```
hom[2, 1, 7];
ut = pol[%, a] /. {a[1] → 1, a[2] → 5}
uT = pol[hom[2, 1, 9], b] /. b[1] → 1

eqs = cfs[comm[uT, ut], u];
coef = Union[vars[ut, a], vars[uT, b]];
eqs = GroebnerBasis[eqs, coef, MonomialOrder → DegreeReverseLexicographic]
sol = Union[Solve[eqs == 0]]
```

Out[54]= $a[4]\,u[0]^2 u[1] + a[3]\,u[1]\,u[2] + 5\,u[0]\,u[3] + u[5]$

Out[55]= $b[8]\,u[0]^3 u[1] + b[7]\,u[1]^3 + b[6]\,u[0]\,u[1]\,u[2] + b[5]\,u[0]^2 u[3] + b[4]\,u[2]\,u[3] + b[3]\,u[1]\,u[4] + b[2]\,u[0]\,u[5] + u[7]$

Out[58]= $\{140 - 7\,b[6] + 14\,b[7] + 6\,b[8],\ -70 + 7\,b[5] - 3\,b[8],\ 105 - 7\,b[4] + 14\,b[7] - 6\,b[8],\ -77 + 7\,b[3] - 7\,b[7] + 3\,b[8],\ -7 + b[2],\ -105 + 49\,a[4] - 15\,b[8],\ -140 + 49\,a[3] - 35\,b[7] + 15\,b[8],\ -980 + 161\,b[8] - 6\,b[8]^2,\ -980 + 56\,b[7] + 105\,b[8] - 6\,b[7]\,b[8],\ -245 + 49\,b[7] - 2\,b[7]^2\}$

Out[59]= $\{\{a[3] \to 5,\ a[4] \to 5,\ b[2] \to 7,\ b[3] \to 14,\ b[4] \to 21,\ b[5] \to 14,\ b[6] \to 42,\ b[7] \to 7,\ b[8] \to \frac{28}{3}\},$
$\{a[3] \to 10,\ a[4] \to \frac{15}{2},\ b[2] \to 7,\ b[3] \to 21,\ b[4] \to 35,\ b[5] \to \frac{35}{2},\ b[6] \to 70,\ b[7] \to \frac{35}{2},\ b[8] \to \frac{35}{2}\},$
$\{a[3] \to \frac{25}{2},\ a[4] \to 5,\ b[2] \to 7,\ b[3] \to \frac{49}{2},\ b[4] \to 42,\ b[5] \to 14,\ b[6] \to 63,\ b[7] \to \frac{35}{2},\ b[8] \to \frac{28}{3}\}\}$

Нашлось 3 ответа!

In[60]:=
```
P1 = ut /. sol[[1]]
P2 = ut /. sol[[2]]
P3 = ut /. sol[[3]]
```

Out[60]= $5\,u[0]^2 u[1] + 5\,u[1]\,u[2] + 5\,u[0]\,u[3] + u[5]$

Out[61]= $\frac{15}{2}\,u[0]^2 u[1] + 10\,u[1]\,u[2] + 5\,u[0]\,u[3] + u[5]$

Out[62]= $5\,u[0]^2 u[1] + \frac{25}{2}\,u[1]\,u[2] + 5\,u[0]\,u[3] + u[5]$

Теперь для каждого из них можно искать симметрии при помощи старой функции sym. Оказывается, то один из ответов это симметрия уравнения порядка 3, то есть КдФ, с точностью до растяжения:

In[63]:=
```
sym[P2, 2, 1, 3]
```

Out[63]= $3\,u[0]\,u[1] + u[3]$

Но два остальных это самостоятельные уравнения. Симметрий порядка 3 у них нет, а есть симметрии порядков 7,11,13,17,... Для них известны представления Лакса; оказывается, что они связаны с линейными уравнениями 3-го порядка (вместо уравнения Шрёдингера для КдФ).

Иерархия Савады-Котеры:

In[64]:=
```
Do[Print[sym[P1, 2, 1, k] /. u[n_] :→ uₙ], {k, 1, 13}]
```

u_1

0

0

0

$5 u_0^2 u_1 + 5 u_1 u_2 + 5 u_0 u_3 + u_5$

0

$\frac{28}{3} u_0^3 u_1 + 7 u_1^3 + 42 u_0 u_1 u_2 + 14 u_0^2 u_3 + 21 u_2 u_3 + 14 u_1 u_4 + 7 u_0 u_5 + u_7$

0

0

0

$\frac{88}{3} u_0^5 u_1 + 330 u_0^2 u_1^3 + \frac{1540}{3} u_0^3 u_1 u_2 + 682 u_1^3 u_2 + 1716 u_0 u_1 u_2^2 + \frac{220}{3} u_0^4 u_3 + 1298 u_0 u_1^2 u_3 + 990 u_0^2 u_2 u_3 + 1320 u_2^2 u_3 + 979 u_1 u_3^2 + 616 u_0^2 u_1 u_4 + 1562 u_1 u_2 u_4 + 880 u_0 u_3 u_4 + \frac{242}{3} u_0^3 u_5 + 396 u_1^2 u_5 + 616 u_0 u_2 u_5 + 286 u_4 u_5 + 286 u_0 u_1 u_6 + 220 u_3 u_6 + 44 u_0^2 u_7 + 121 u_2 u_7 + 44 u_1 u_8 + 11 u_0 u_9 + u_{11}$

0

$\frac{455}{9} u_0^6 u_1 + \frac{3640}{3} u_0^3 u_1^3 + 546 u_1^5 + 1365 u_0^4 u_1 u_2 + 8736 u_0 u_1^3 u_2 + 9919 u_0^2 u_1 u_2^2 + \frac{27\,859}{3} u_1 u_2^3 + \frac{455}{3} u_0^5 u_3 + 7462 u_0^2 u_1^2 u_3 + 3640 u_0^3 u_2 u_3 + 20\,826 u_1^2 u_2 u_3 + 15\,925 u_0 u_2^2 u_3 + 11\,934 u_0 u_1 u_3^2 + 3640 u_3^3 + \frac{6734}{3} u_0^3 u_1 u_4 + 4160 u_1^3 u_4 + 19\,266 u_0 u_1 u_2 u_4 + 5278 u_0^2 u_3 u_4 + 17\,472 u_2 u_3 u_4 + 5174 u_1 u_4^2 + \frac{637}{3} u_0^4 u_5 + 4940 u_0 u_1^2 u_5 + 3640 u_0^2 u_2 u_5 + 5915 u_2^2 u_5 + 8762 u_1 u_3 u_5 + 3822 u_0 u_4 u_5 + 1651 u_0^2 u_1 u_6 + 5187 u_1 u_2 u_6 + 2847 u_0 u_3 u_6 + 1079 u_5 u_6 + \frac{481}{3} u_0^3 u_7 + 1014 u_1^2 u_7 + 1521 u_0 u_2 u_7 + 845 u_4 u_7 + 546 u_0 u_1 u_8 + 507 u_3 u_8 + 65 u_0^2 u_9 + 221 u_2 u_9 + 65 u_1 u_{10} + 13 u_0 u_{11} + u_{13}$

Иерархия Каупа-Купершмидта:

In[65]:=
```
Do[Print[sym[P3, 2, 1, k] /. u[n_] :> u_n], {k, 1, 13}]
```

u_1

0

0

0

$5 u_0^2 u_1 + \frac{25 u_1 u_2}{2} + 5 u_0 u_3 + u_5$

0

$\frac{28}{3} u_0^3 u_1 + \frac{35 u_1^3}{2} + 63 u_0 u_1 u_2 + 14 u_0^2 u_3 + 42 u_2 u_3 + \frac{49 u_1 u_4}{2} + 7 u_0 u_5 + u_7$

0

0

0

$$\frac{88}{3} u_0^5 u_1 + 495 u_0^2 u_1^3 + \frac{1870}{3} u_0^3 u_1 u_2 + \frac{3179}{2} u_1^3 u_2 + 2871 u_0 u_1 u_2^2 + \frac{220}{3} u_0^4 u_3 + 2123 u_0 u_1^2 u_3 + 1320 u_0^2 u_2 u_3 + 2310 u_2^2 u_3 + 1771 u_1 u_3^2 + 781 u_0^2 u_1 u_4 + \frac{5731}{2} u_1 u_2 u_4 + 1375 u_0 u_3 u_4 + \frac{242}{3} u_0^3 u_5 + 726 u_1^2 u_5 + 913 u_0 u_2 u_5 + 550 u_4 u_5 + 385 u_0 u_1 u_6 + 385 u_3 u_6 + 44 u_0^2 u_7 + 187 u_2 u_7 + \frac{121 u_1 u_8}{2} + 11 u_0 u_9 + u_{11}$$

0

$$\frac{455}{9} u_0^6 u_1 + \frac{5005}{3} u_0^3 u_1^3 + \frac{21021 u_1^5}{16} + \frac{3185}{2} u_0^4 u_1 u_2 + \frac{63609}{4} u_0 u_1^3 u_2 + \frac{29393}{2} u_0^2 u_1 u_2^2 + \frac{444821}{24} u_1 u_2^3 + \frac{455}{3} u_0^5 u_3 + \frac{21749}{2} u_0^2 u_1^2 u_3 + 4550 u_0^3 u_2 u_3 + \frac{83421}{2} u_1^2 u_2 u_3 + \frac{103285}{4} u_0 u_2^2 u_3 + \frac{77493}{4} u_0 u_1 u_3^2 + 6370 u_3^3 + \frac{8099}{3} u_0^3 u_1 u_4 + \frac{66859}{8} u_1^3 u_4 + \frac{124293}{4} u_0 u_1 u_2 u_4 + \frac{14651}{2} u_0^2 u_3 u_4 + 30849 u_2 u_3 u_4 + \frac{18967}{2} u_1 u_4^2 + \frac{637}{3} u_0^4 u_5 + \frac{30953}{4} u_0 u_1^2 u_5 + \frac{9737}{2} u_0^2 u_2 u_5 + \frac{41405}{4} u_2^2 u_5 + \frac{63089}{4} u_1 u_3 u_5 + \frac{12285}{2} u_0 u_4 u_5 + \frac{4121}{2} u_0^2 u_1 u_6 + \frac{35997}{4} u_1 u_2 u_6 + \frac{8697}{2} u_0 u_3 u_6 + \frac{4069 u_5 u_6}{2} + \frac{481}{3} u_0^3 u_7 + \frac{6747}{4} u_1^2 u_7 + 2145 u_0 u_2 u_7 + 1508 u_4 u_7 + 702 u_0 u_1 u_8 + 819 u_3 u_8 + 65 u_0^2 u_9 + \frac{637 u_2 u_9}{2} + \frac{169 u_1 u_{10}}{2} + 13 u_0 u_{11} + u_{13}$$

4.2 вес $w(u_j) = 1 + 2j$

Этот случай сложнее для анализа, так как здесь число мономов заданного веса значительно больше. Однако, ответ тут оказывается не такой интересный. Обнаруживается лишь одно новое уравнение, порядка 3, обладающее симметрией порядка 5 (в отличие от уравнений из предыдущего пункта, оно линеаризуемое, как и уравнение Бюргерса, хотя и более сложной подстановкой). Сразу перейдем к нему, пропуская другие случаи.

In[66]:=
```
ut = pol[hom[1, 2, 7], a] /. {a[1] → 1, a[2] → 3}
uT = pol[hom[1, 2, 11], b] /. b[1] → 1

eqs = cfs[comm[uT, ut], u];
coef = Union[vars[ut, a], vars[uT, b]];
eqs = GroebnerBasis[eqs, coef, MonomialOrder → DegreeReverseLexicographic];
sol = Union[Solve[eqs == 0]]
```

Out[66]= `a[5] u[0]^7 + a[4] u[0]^4 u[1] + a[3] u[0] u[1]^2 + 3 u[0]^2 u[2] + u[3]`

Out[67]= `b[12] u[0]^11 + b[11] u[0]^8 u[1] + b[10] u[0]^5 u[1]^2 + b[8] u[0]^2 u[1]^3 + b[9] u[0]^6 u[2] + b[7] u[0]^3 u[1] u[2] + b[5] u[1]^2 u[2] + b[4] u[0] u[2]^2 + b[6] u[0]^4 u[3] + b[3] u[0] u[1] u[3] + b[2] u[0]^2 u[4] + u[5]`

Out[71]= `{{a[3] → 9, a[4] → 3, a[5] → 0, b[2] → 5, b[3] → 40, b[4] → 25, b[5] → 50, b[6] → 10, b[7] → 120, b[8] → 140, b[9] → 10, b[10] → 70, b[11] → 5, b[12] → 0}}`

Иерархия Ибрагимова-Шабата:

In[72]:=
```
P = ut /. sol[[1]]
Do[Print[sym[P, 1, 2, k] /. u[n_] :→ u_n], {k, 1, 9}]
```

Out[72]= `3 u[0]^4 u[1] + 9 u[0] u[1]^2 + 3 u[0]^2 u[2] + u[3]`

u_1

0

$3 u_0^4 u_1 + 9 u_0 u_1^2 + 3 u_0^2 u_2 + u_3$

0

$5 u_0^8 u_1 + 70 u_0^5 u_1^2 + 140 u_0^2 u_1^3 + 10 u_0^6 u_2 + 120 u_0^3 u_1 u_2 + 50 u_1^2 u_2 + 25 u_0 u_2^2 + 10 u_0^4 u_3 + 40 u_0 u_1 u_3 + 5 u_0^2 u_4 + u_5$

0

$$7 u_0^{12} u_1 + 231 u_0^9 u_1^2 + 1876 u_0^6 u_1^3 + 3360 u_0^3 u_1^4 + 280 u_1^5 + 21 u_0^{10} u_2 + 784 u_0^7 u_1 u_2 + 4914 u_0^4 u_1^2 u_2 + 3164 u_0 u_1^3 u_2 + 483 u_0^5 u_2^2 + 2422 u_0^2 u_1 u_2^2 + 126 u_2^3 + 35 u_0^8 u_3 + 728 u_0^5 u_1 u_3 + 1848 u_0^2 u_1^2 u_3 + 588 u_0^3 u_2 u_3 + 518 u_1 u_2 u_3 + 91 u_0 u_3^2 + 35 u_0^6 u_4 + 350 u_0^3 u_1 u_4 + 140 u_1^2 u_4 + 140 u_0 u_2 u_4 + 21 u_0^4 u_5 + 70 u_0 u_1 u_5 + 7 u_0^2 u_6 + u_7$$

0

$$9 u_0^{16} u_1 + 540 u_0^{13} u_1^2 + 9864 u_0^{10} u_1^3 + 62\,208 u_0^7 u_1^4 + 110\,736 u_0^4 u_1^5 + 22\,896 u_0 u_1^6 + 36 u_0^{14} u_2 + 2736 u_0^{11} u_1 u_2 + 48\,636 u_0^8 u_1^2 u_2 + 231\,408 u_0^5 u_1^3 u_2 + 190\,224 u_0^2 u_1^4 u_2 + 2790 u_0^9 u_2^2 + 60\,408 u_0^6 u_1 u_2^2 + 201\,168 u_0^3 u_1^2 u_2^2 + 32\,508 u_1^3 u_2^2 + 17\,424 u_0^4 u_2^3 + 34\,776 u_0 u_1 u_2^3 + 84 u_0^{12} u_3 + 4176 u_0^9 u_1 u_3 + 45\,120 u_0^6 u_1^2 u_3 + 99\,936 u_0^3 u_1^3 u_3 + 11\,424 u_1^4 u_3 + 6672 u_0^7 u_2 u_3 + 77\,400 u_0^4 u_1 u_2 u_3 + 75\,048 u_0 u_1^2 u_2 u_3 + 19\,416 u_0^2 u_2^2 u_3 + 2964 u_0^5 u_3^2 + 14\,268 u_0^2 u_1 u_3^2 + 2400 u_2 u_3^2 + 126 u_0^{10} u_4 + 3960 u_0^7 u_1 u_4 + 22\,680 u_0^4 u_1^2 u_4 + 14\,040 u_0 u_1^3 u_4 + 4680 u_0^5 u_2 u_4 + 22\,140 u_0^2 u_1 u_2 u_4 + 1836 u_2^2 u_4 + 2760 u_0^3 u_3 u_4 + 2478 u_1 u_3 u_4 + 321 u_0 u_4^2 + 126 u_0^8 u_5 + 2304 u_0^5 u_1 u_5 + 5328 u_0^2 u_1^2 u_5 + 1800 u_0^3 u_2 u_5 + 1566 u_1 u_2 u_5 + 534 u_0 u_3 u_5 + 84 u_0^6 u_6 + 756 u_0^3 u_1 u_6 + 294 u_1^2 u_6 + 294 u_0 u_2 u_6 + 36 u_0^4 u_7 + 108 u_0 u_1 u_7 + 9 u_0^2 u_8 + u_9$$